Calculus Single Variable

Brian E. Blank and Steven G. Krantz

Section 9.1
Infinite Series - Introduction

1. Partial Sums

Given a sequence $\{ a_n, n = 1 .. \infty \}$, we can form an associated sequence of partial sums $\{ S_N, N = 1 .. \infty \}$ with S_N defined by

$$S_N = \sum_{n=1}^{N} a_n .$$

We perform the same construction even if the sequence $\{ a_n \}$ does not begin with the index $n = 1$. For example, the first three partial sums of the sequence $\{ \frac{1}{n-\pi}, n = 4 .. \infty \}$ are $S_1 = \frac{1}{4-\pi}$, $S_2 = \frac{1}{4-\pi} + \frac{1}{5-\pi}$, and $S_3 = \frac{1}{4-\pi} + \frac{1}{5-\pi} + \frac{1}{6-\pi}$.

A ***geometric sequence*** $\ldots\ a_{n-1},\ a_n,\ a_{n+1},\ \ldots$ is one in which there is a single value r equal to the all the ratios $\ldots,\ \frac{a_n}{a_{n-1}},\ \frac{a_{n+1}}{a_n},\ \ldots$ of successive terms. If a is the first term and r is the ratio, then the geometric sequence is

$$a,\ a\,r,\ a\,r^2,\ a\,r^3,\ a\,r^4,\ \ldots .$$

A ***geometric sum*** or ***geometric series*** is the sum of consecutive terms of a geometric sequence. Thus, if M and N are integers with $M < N$, then

$$\sum_{n=M}^{N} a\, r^n$$

is a geometric series.

Example 1.1: Geometric Series

Let a and r be constants with $r \neq 1$. Calculate the N^{th} partial sum of $\{ a\, r^n, n = 0 .. \infty \}$,

Solution:

The sequence is a, $a\, r$, $a\, r^2$, $a\, r^3$, $a\, r^4$, ...

The first term is a, the second term is $a\, r$, the third term is $a\, r^2$, and, in general, the power of r is one less than the index of the term. Thus, the N^{th} term is $a\, r^{(N-1)}$.

Adding the first N terms and factoring the a, we have

$$S_N = a\,(1 + r + r^2 + r^3 + \;...\; + r^{(N-1)}).$$

Notice that

$$r\, S_N = a\,(r + r^2 + r^3 + \;...\; + r^{(N-1)} + r^N).$$

It follows that $S_N - r\, S_N = a\,(1 - r^N)$, or $S_N = \dfrac{a\,(1 - r^N)}{1 - r}$.

In other words,

$$\sum_{n=0}^{N-1} a\, r^n = \frac{a\,(1 - r^N)}{1 - r}.$$

Ω

It is convenient to have a generalization of the formula found in Example 1.1. First, let $L = N - 1$ and $k = n$:

$$\sum_{k=0}^{L} a\, r^k = \frac{a\,(1 - r^{(L+1)})}{1 - r}.$$

Next, for any integer M, multiply both sides by r^M:

$$\sum_{k=0}^{L} a\, r^{(k+M)} = \frac{a\, r^M\,(1 - r^{(L+1)})}{1 - r}.$$

Finally, let $N = L + M$ and change the index of summation for the series to n where $n = k + M$. Since the initial value of n is M and the final value is N, we have, for integers M and N with $M < N$:

General Formula For a Geometric Sum: $\displaystyle\sum_{n=M}^{N} a\, r^n = \frac{a\, r^M\,(1 - r^{(N-M+1)})}{1 - r} = \frac{a\,(r^M - r^{(N+1)})}{1 - r}$.

Example 1.2: Calculate $\frac{5}{9} + \frac{5}{27} + \frac{5}{81} + \frac{5}{243} + \frac{5}{729}$.

Solution:
We can add this up in the normal way:

```
> 5/9+5/27+5/81+5/243+5/729;
```

$$\frac{605}{729}$$

Or, we can use the formula derived before this example with $a = 5$, $r = \frac{1}{3}$, $M = 2$, and $N = 6$:

```
> formula := Sum(a*r^n, n = M .. N) = a*(r^M-r^(N+1))/(1-r);
```

$$formula := \sum_{n=M}^{N} a\, r^n = \frac{a\,(r^M - r^{(N+1)})}{1-r}$$

```
> subs( {a=5, r=1/3, M=2, N=6} , formula);
```

$$\sum_{n=2}^{6} \left(5\left(\frac{1}{3}\right)^n\right) = \frac{605}{729}$$

Ω

2. Partial Sums of Collapsing or Telescoping Sums

If f is a function and $a_n = \mathrm{f}(n) - \mathrm{f}(n+1)$, then we say that $\sum_{n=1}^{N} a_n$ is a ***collapsing***, or ***telescoping***, series.

Notice that the cancellation that occurs when consecutive terms are added:

$$a_n + a_{n+1} = \big(\mathrm{f}(n) - \mathrm{f}(n+1)\big) + \big(\mathrm{f}(n+1) - \mathrm{f}(n+2)\big) = \mathrm{f}(n) - \mathrm{f}(n+2).$$

So,

$$\sum_{n=1}^{N} a_n = \big(\mathrm{f}(1) - \mathrm{f}(2)\big) + \big(\mathrm{f}(2) - \mathrm{f}(3)\big) + \big(\mathrm{f}(3) - \mathrm{f}(4)\big) + \ldots + \big(\mathrm{f}(N-1) - \mathrm{f}(N)\big) + \big(\mathrm{f}(N) - \mathrm{f}(N+1)\big)$$

simplifies to $\mathrm{f}(1) - \mathrm{f}(N+1)$.

Thus,

$$\sum_{n=1}^{N} (\mathrm{f}(n) - \mathrm{f}(n+1)) = \mathrm{f}(1) - \mathrm{f}(N+1).$$

Example 2.1: Calculate the 342^{th} partial sum of $\{\frac{1}{n\,(n+1)}, n = 1 .. \infty\}$.

Solution: Notice that $\frac{1}{n\,(n+1)} = \frac{1}{n} - \frac{1}{n+1}$. Using the collapsing sum formula with $f(n) = \frac{1}{n}$, we have

$$S_{342} = \sum_{n=1}^{342} \frac{1}{n\,(n+1)} = \sum_{n=1}^{342} (f(n) - f(n+1)) = f(1) - f(343) = 1 - \frac{1}{343} = \frac{342}{343}.$$

Here is a verification with **Maple**:

```
> Sum(1/n/(n+1), n = 1 .. 342) = sum(1/n/(n+1), n = 1 .. 342);
```

$$\sum_{n=1}^{342} \frac{1}{n\,(n+1)} = \frac{342}{343}$$

(Sometimes **Maple** does calculations like this by brute force. In this case, **Maple** knows the general simplification:

```
> Sum(1/n/(n+1), n = 1 .. N) = sum(1/n/(n+1), n = 1 .. N);
```

$$\sum_{n=1}^{N} \frac{1}{n\,(n+1)} = -\frac{1}{N+1} + 1$$

Ω

3. Infinite Series

Given a sequence $\{a_n, n = 1 .. \infty\}$ with its associated sequence of partial sums $\{S_N, N = 1 .. \infty\}$, we define the infinite series $\sum_{n=1}^{\infty} a_n$ by

$$\sum_{n=1}^{\infty} a_n = \lim_{N \to \infty} S_N$$

provided the limit exists**.** In this case we say the infinite series is ***convergent*****.** Otherwise we say it is ***divergent*****.**

Example 3.1:

Estimate the partial sums of the series $\sum_{n=1}^{\infty} \frac{n}{2n+1}$. Show that the series diverges.

Solution:

The N^{th} partial sum is $S_N = \frac{1}{3} + \frac{2}{5} + \frac{3}{7} + \ldots + \frac{N-1}{2N-1} + \frac{N}{2N+1}$.

To get some sense of these sums we calculate several:

```
> seq(Sum(n/(2*n+1), n = 1..N) = evalf(add(n/(2*n+1), n = 1..N)),
  N = 1 .. 10);
```

$$\sum_{n=1}^{1} \frac{n}{2n+1} = 0.3333333333, \sum_{n=1}^{2} \frac{n}{2n+1} = 0.7333333333, \sum_{n=1}^{3} \frac{n}{2n+1} = 1.161904762,$$
$$\sum_{n=1}^{4} \frac{n}{2n+1} = 1.606349206, \sum_{n=1}^{5} \frac{n}{2n+1} = 2.060894661, \sum_{n=1}^{6} \frac{n}{2n+1} = 2.522433122,$$
$$\sum_{n=1}^{7} \frac{n}{2n+1} = 2.989099789, \sum_{n=1}^{8} \frac{n}{2n+1} = 3.459688024, \sum_{n=1}^{9} \frac{n}{2n+1} = 3.933372235,$$
$$\sum_{n=1}^{10} \frac{n}{2n+1} = 4.409562711$$

It appears that $\frac{N}{3} \le S_N$. If this were true we could conclude that $\lim_{N \to \infty} S_N = \infty$, which would show that the series is divergent. Upon closer observation we see that *every* term of the series is no smaller than 1/3. We prove this as follows. Adding $2n$ to each side of the inequality $1 \le n$ results in $2n + 1 \le 3n$. Dividing each side of the last inequality by $3(2n+1)$ gives us the required inequality $\frac{1}{3} \le \frac{n}{2n+1}$. Thus,

$$\frac{N}{3} = \sum_{n=1}^{N} \frac{1}{3} \le \sum_{n=1}^{N} \frac{n}{2n+1} = S_N$$

for every N, which, as we have seen, implies the divergence of the given series.

Ω

4. Collapsing or Telescoping Series

Theorem:

If $\lim_{x \to \infty} f(x) = L$, then $\sum_{n=1}^{\infty} (f(n) - f(n+1)) = f(1) - L$.

Proof:

The series $\sum_{n=1}^{\infty} (f(n) - f(n+1))$ telescopes. We have seen that its N^{th} partial sum S_N is given by $S_N = f(1) - f(N+1)$. Therefore

$$\sum_{n=1}^{\infty} (f(n) - f(n+1)) = \lim_{N \to \infty} S_N = \lim_{N \to \infty} (f(1) - f(N+1)) = f(1) - L.$$

Example 4.1: Show that the series $\sum_{n=1}^{\infty} \frac{1}{n(n+1)}$ is convergent. What is the value of the series?

Solution: As we have seen, $\sum_{n=1}^{\infty} \frac{1}{n(n+1)}$ is the collapsing series $\sum_{n=1}^{\infty} (f(n) - f(n+1))$ with $f(n) = \frac{1}{n}$.

The N^{th} partial sum S_N is given by

$$S_N = f(1) - f(N+1) = 1 - \frac{1}{N+1}.$$

Clearly

$$\lim_{N \to \infty} S_N = 1.$$

Thus, $\sum_{n=1}^{\infty} \frac{1}{n(n+1)}$ converges and $\sum_{n=1}^{\infty} \frac{1}{n(n+1)} = 1.$

Ω

Example 4.2:

Calculate the partial sums of the series $\sum_{n=1}^{\infty} \frac{n}{(n+1)!}$. Evaluate the sum of the series.

Solution:

The N^{th} partial sum is $S_N = \frac{1}{2!} + \frac{2}{3!} + \frac{3}{4!} + \dots + \frac{N-1}{N!} + \frac{N}{(N+1)!}$.

Notice that the n^{th} term of the series satisfies

$$\frac{n}{(n+1)!} = \frac{(n+1)-1}{(n+1)!} = \frac{1}{n!} - \frac{1}{(n+1)!} = f(n) - f(n+1)$$

where $f(n) = \frac{1}{n!}$. It follows that $S_N = f(1) - f(N+1) = \frac{1}{1!} - \frac{1}{(N+1)!}$ and

$$\sum_{n=1}^{\infty} \frac{n}{(n+1)!} = \lim_{N \to \infty} S_N = \lim_{N \to \infty} \left(1 - \frac{1}{(N+1)!}\right) = 1.$$

Ω

5. Geometric Series

Let a and r be constants with $r \neq 1$. The series,

$$a + a\,r + a\,r^2 + a\,r^3 + \dots ,$$

that is, $\sum_{n=0}^{\infty} a\,r^n$, is an ***(infinite) geometric series***. We have seen that its N^{th} partial sum S_N is given by

$$S_N = \frac{a\,(1 - r^N)}{1 - r}.$$

Notice that $\lim_{N \to \infty} S_N$ exists if and only if $|r| < 1$. In this case $\lim_{N \to \infty} S_N = \frac{a}{1 - r}$. In summary,

$$\sum_{n=0}^{\infty} a\,r^n \quad \textit{diverges} \text{ if } \quad 1 \leq |r|$$

and

$$\sum_{n=0}^{\infty} a\,r^n = \frac{a}{1 - r} \qquad \text{if} \quad |r| < 1.$$

A Generalization:

By factoring we see that $\sum_{n=M}^{\infty} a\,r^n = r^M \left(\sum_{n=0}^{\infty} a\,r^n \right)$. Therefore, we have

$$\sum_{n=M}^{\infty} a\, r^n = \frac{a\, r^M}{1-r}.$$

6. Expressing a Repeating Decimal as a Rational Number

Exercise 20, page 628

Express **0.98398398398...** as a rational number $\frac{m}{n}$ with m and n integers.

Solution: We have

$$0.98398398398\,,,, \;=\; 983\left(\frac{1}{10^3}+\frac{1}{10^6}+\frac{1}{10^9}+\ldots\right).$$

.

We apply the formula $\sum_{n=M}^{\infty} a\, r^n = \frac{a\, r^M}{1-r}$ with $a = 1$, $M = 1$, and $r = \frac{1}{10^3}$.

We find $\frac{1}{10^3}+\frac{1}{10^6}+\frac{1}{10^9}+\ldots = \frac{1}{10^3\left(1-\frac{1}{10^3}\right)} = \frac{1}{999}$. Thus,

$$0.98398398398\,,,, \;=\; 983\left(\frac{1}{999}\right) = \frac{983}{999}.$$

It would be only a little bit of extra work if the repetition did not begin right after the decimal.

Example:

Express **0.97398398398...** as a rational number $\frac{m}{n}$ with m and n integers.

Solution: We have

$$0.97398398398\ ,,, = \frac{97}{100} + \frac{398}{10^5} + \frac{398}{10^8} + \frac{398}{10^{11}} + \frac{398}{10^{14}} + \cdots = \frac{97}{100} + \frac{398}{10^5}\left(1 + \frac{1}{10^3} + \frac{1}{10^6} + \frac{1}{10^9} + \cdots\right).$$

.

We apply the formula $\sum_{n=M}^{\infty} a\,r^n = \frac{a\,r^M}{1-r}$ with $a = 1$, $M = 0$, and $r = \frac{1}{10^3}$.

We find $1 + \frac{1}{10^3} + \frac{1}{10^6} + \frac{1}{10^9} + \cdots = \frac{1}{1 - \frac{1}{10^3}} = \frac{1000}{999}$. Thus,

$$0.97398398398\ ,,, = \frac{97}{100} + \frac{398}{10^5}\left(\frac{1000}{999}\right) = \frac{97301}{99900}.$$

As a check, we ask **Maple** for a 15 decimal place numerical evaluation of our answer:

```
> evalf(97301/99900, 15);
```

0.973983983983984

7. The Multiplier Effect (in Economics)

Exercise 43, page 629

Suppose that every dollar that we spend gives rise (through wages, profits, etc.) to 90 cents for someone else to spend. That 90 cents will generate a further 81 cents for spending, and so on. How much spending will result from the purchase of a \$16000 automobile? This phenomenon is known as the multiplier effect.

The spending, including the initial purchase of the car, is

$$16000 + \left(16000\,\frac{9}{10}\right) + \left(16000\,\frac{9}{10}\right)\left(\frac{9}{10}\right) + \left(\left(16000\,\frac{9}{10}\right)\left(\frac{9}{10}\right)\right)\left(\frac{9}{10}\right) +$$

• • •

or

```
> Sum(16000*(9/10)^k, k=0..infinity) = sum(16000*(9/10)^k,
  k=0..infinity);
```

$$\sum_{k=0}^{\infty} \left(16000 \left(\frac{9}{10}\right)^k\right) = 160000$$

8. Two Trains and A Fly

An eastbound and a westbound train head toward each other on the same track. The speed of each train is 120 miles per hour. When the trains are 120 miles apart, a fly departs the front of the eastbound train and flies to the front of the westbound train, whereupon the fly reverses direction and flies back to the eastbound train. If the speed of the fly is 90 miles per hour and the fly repeats this process until the two trains collide, how far does the fly fly?

Let d_n denote the distance between the two trains when the fly begins its n^{th} trip. This is the distance the fly and the oncoming train cover during the n^{th} trip. We are given $d_1 = 120$ miles. Let f_n denote the distance the fly travels in its n^{th} trip between the two trains. During each trip the fly and the oncoming train approach each other at 90 + 60, or 150, miles per hour. Since the speed of the fly, namely 90, is 3/5 of the speed at which the fly and the oncoming train near each other, namely 150, we have $f_n = \frac{3\, d_n}{5}$. Also, the time t_n of the n^{th} trip is given by $t_n = \frac{d_n}{150}$ hours. During the n^{th} trip the two trains approach each other at 60 + 60, or 120, miles per hour. The distance by which they close on each other during the n^{th} trip is therefore $\frac{120\, d_n}{150}$, or $\frac{4\, d_n}{5}$, miles. Thus, we have $d_{n+1} = d_n - \frac{4\, d_n}{5}$, or $d_{n+1} = \frac{d_n}{5}$. It follows that $d_2 = \frac{120}{5}$, $d_3 = \frac{d_2}{5} =$

$\frac{120}{5^2}$, and, in general, $d_n = \frac{120}{5^{(n-1)}}$. Thus, $f_n = \frac{3\,d_n}{5} = \frac{360}{5^n}$. The distance the fly flies is

$$\sum_{n=1}^{\infty} f_n = \sum_{n=1}^{\infty} \frac{360}{5^n} = 90.$$

A simpler way to solve this problem is to notice that, since the two trains approach each other at 120 miles per hour and since they are 120 miles apart when the fly sets off, they will collide one hour later. At a constant speed of 90 miles per hour, the fly flies 90 miles in this time. When the Hungarian-American methematician John von Neumann, famous for his computational abilities among other things, was asked this problem, he responded with the correct answer almost instantaneously. The poser of the problem, deprived of his amusement, said "You must know the trick." Replied von Neumann, "I just summed the series."

9. The Harmonic Series

The so-called ***harmonic series*** $\sum_{n=1}^{\infty} \frac{1}{n}$ is known to diverge, as we will soon see for ourselves.

The N^{th} partial sum of the harmonic series is usually denoted by H_N:

$$H_N = \sum_{n=1}^{N} \frac{1}{n}.$$

Figure 1 below shows how H_N can be interpreted geometrically as the area of rectangles each having base 1.



Figure 1

The first rectangle has height 1/1. Its base is the interval [1,2]. It therefore has area 1. The second rectangle has height 1/2. Its base is the interval [2,3], which has length 1. The area of the second rectangle is therefore 1/2. In general, the height of the n^{th} rectangle is 1/n and the base is the interval [n , n+1], which has length 1. The area of the n^{th} rectangle is therefore 1/n.

We see that

$$\ln(N+1) = \int_1^{N+1} \frac{1}{x}\,dx < \sum_{n=1}^{N} \frac{1}{n} = H_N .$$

It follows that $\lim_{N \to \infty} H_N = \infty$, which shows that the harmonic series $\sum_{n=1}^{\infty} \frac{1}{n}$ diverges.

Example 9.1: (Exercise 40, page 629)

Use the inequality $\frac{x}{2} < \ln(1+x)$ for $0 < x < 1$ to show that $\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right)$ is divergent.

Solution:

First of all, let us get a sense of this series by calculating several of the partial sums. Here are the first 25:

```
> seq(Sum(ln(1+1/n),n = 1..infinity) = add(evalf(ln(1+1/n)), n = 1..N), N=1..25);
```

$$\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 0.6931471806, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 1.098612289, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 1.386294361,$$

$$\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 1.609437912, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 1.791759469, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 1.945910149,$$

$$\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.079441542, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.197224578, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.302585094,$$

$$\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.397895274, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.484906651, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.564949358,$$

$$\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.639057330, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.708050201, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.772588722,$$

$$\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.833213344, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.890371757, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.944438979,$$

$$\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 2.995732273, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 3.044522437, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 3.091042453,$$

$$\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 3.135494215, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 3.178053830, \sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 3.218875825,$$

$$\sum_{n=1}^{\infty} \ln\left(1+\frac{1}{n}\right) = 3.258096538$$

The partial sums increase because each new term added is positive. However, the rate at which the partial sums increase is decreasing. The numerical evidence for convergence or divergence is not especially clear. The analytic approach we now take will be convincing.

The following graph shows that $\frac{x}{2} < \ln(1+x)$ for $0 < x < 1$:

```
> plot([ln(1+x),x/2], x = 0 .. 1, color = [NAVY,SIENNA],
  thickness=2);
```



We can derive the inequality $\frac{x}{2} < \ln(1+x)$ for $0 < x < 1$ by noticing that $1+t < 2$ for $0 < t < 1$ and therefore

$\frac{1}{2} < \frac{1}{1+t}$ for $0 < t < 1$. We integrate each side of the last inequality over the interval $[0, x]$, $0 < x < 1$, to obtain

```
> int(1/2, t = 0 ..x) < int(1/(1+t), t = 0 .. x);
```

$$\frac{x}{2} < \ln(1+x)$$

It follows that

```
> Sum(ln(1+1/n), n = 1 .. N) > Sum(1/2/n, n = 1 .. N);
```

$$\sum_{n=1}^{N} \left(\frac{1}{2\,n}\right) < \sum_{n=1}^{N} \ln\left(1+\frac{1}{n}\right)$$

or

```
> Sum(ln(1+1/n), n = 1 .. N) > (1/2)*Sum(1/n, n = 1 .. N);
```

$$\frac{1}{2}\left(\sum_{n=1}^{N} \frac{1}{n}\right) < \sum_{n=1}^{N} \ln\left(1+\frac{1}{n}\right)$$

Thus, $\frac{H_N}{2} < \sum_{n=1}^{N} \ln\left(1+\frac{1}{n}\right)$. Since the partial sums H_N of the harmonic series tend to infinity, we conclude the same for the partial sums of the given series. It therefore diverges.

Ω

Example 9.2: (Exercise 49, page 629)

Homer Woodman has a *large* supply of wooden dominoes. They are 1 inch long and 3/16 inch thick.

He stacks N of them so that, for each $2 \le n \le N$, the n^{th} domino, counting from the bottom of the stack, protrudes

$$\frac{1}{2\,(N-n+1)} \quad \text{inch}$$

over the end of the $(n-1)^{st}$ domino. See Figure 2.



Figure 2: Mr Woodman's Tower of Dominoes

Show that the center of mass of Mr. Woodman's tower of dominoes lies over the bottom domino (and so the stack will not fall). Deduce that by using a sufficiently large number N, Mr. Woodman can make his stack span, from left to right, any given distance. About how many dominoes would it take to span the 10 foot length of his playroom. How high would the tower be?

Solution:

Position the first domino so that when viewed from the side, its upper right vertex (x_1, y_1) is at $(1 , 3/16)$ in the vertical
xy-plane.. Mr. Woodman places the second domino so that its upper right vertex (x_2, y_2) is at $\left(x_1 + \frac{1}{2(N-1)} , y_1 + \frac{3}{16} \right)$, or $\left(1 + \frac{1}{2(N-1)} , \frac{2(3)}{16} \right)$. Mr. Woodman places the third domino so that its upper right vertex (x_3, y_3) is at $\left(x_2 + \frac{1}{2(N-2)} , y_2 + \frac{3}{16} \right)$, or $\left(1 + \frac{1}{2(N-1)} + \frac{1}{2(N-2)} , \frac{3(3)}{16} \right)$. In general, the n^{th} domino has upper right vertex

$$\left(1 + \left(\sum_{k=1}^{n-1} \frac{1}{2(N-k)} \right) , n\left(\frac{3}{16} \right) \right)$$

Let ξ_N be the *x*-coordinate of the center of mass of the N dominoes. Since the area of each domino is 3/16, assuming constant mass density δ, we calculate that the mass is $\frac{3\,\delta\,N}{16}$ and

$$\xi_N = \frac{1}{\frac{3\,\delta\,N}{16}} M_{x=0}$$

where

$$M_{x=0} = \sum_{n=1}^{N} \int_{x_n - 1}^{x_n} \frac{x\,\delta\,3}{16}\,dx = \frac{\delta\,3\left(\sum_{n=1}^{N} (x_n^{\,2} - (x_n - 1)^2) \right)}{32} = \frac{\delta\,3\left(\sum_{n=1}^{N} (x_n^{\,2} - (x_n - 1)^2) \right)}{16}$$

$$= \frac{\delta\,3\left(\sum_{n=1}^{N} (2\,x_n - 1) \right)}{32}.$$

Thus,

$$\xi_N = \frac{1}{2N} \sum_{n=1}^{N} (2\,x_n - 1) = \frac{1}{N} \sum_{n=1}^{N} x_n - \frac{1}{2} = \frac{1}{N}\left(\sum_{n=1}^{N} \left(1 + \left(\sum_{k=1}^{n-1} \frac{1}{2(N-k)} \right) \right) \right) - \frac{1}{2}$$

$$= \frac{1}{N}\left(\sum_{k=1}^{N-1}\left(\sum_{n=k+1}^{N}\frac{1}{2(N-k)}\right)\right)+\frac{1}{2} = \frac{1}{N}\left(\sum_{k=1}^{N-1}\frac{N-k}{2(N-k)}\right)+\frac{1}{2} = \frac{1}{N}\left(\sum_{k=1}^{N-1}\frac{1}{2}\right)+\frac{1}{2}$$

$$= \frac{2N-1}{2N}.$$

Notice that $\xi_N < 1$. The domino tower therefore does not topple.

To determine how many dominoes are required to span 10 feet, or 120 inches, we solve

```
> eqn := 1 + sum(1/2/(N-k) , k = 1 .. N-1) = 120;
```

$$eqn := 1+\left(\sum_{k=1}^{N-1}\left(\frac{1}{2(N-k)}\right)\right)=120$$

This equation is equivalent to

```
> eqn2 := sum(1/(N-k) , k = 1 .. N-1) = 238;
```

$$eqn2 := \sum_{k=1}^{N-1}\frac{1}{N-k}=238$$

or, setting $n = N - k$,

```
> eqn3 := sum(1/n , n = 1 .. N-1) = 238;
```

$$eqn3 := \Psi(N)+\gamma=238$$

(**Maple** can express the partial sums of the harmonic series in terms of the Psi function and the Euler-Mascheroni constant γ.)

```
> N = fsolve(eqn3, N,10^103 .. 10^104);
```

$$N = 0.1292424300\ 10^{104}$$

The structure is therefore $3\left(12924243\times 10^{96}\right) /16$ inches high, or $3.824645775\times 10^{97}$ miles.

```
> (.1292424300e104)*3/15/12/5280;
```

$$0.4079622159\ 10^{98}$$

miles high.

Ω



10. Other Exercises from the Text

Exercise 51 (page 630)

Plot the partial sums of $\sum_{n=1}^{\infty} \frac{e^{(-n)}}{n}$. From your plot does it appear that the given series converges? If so, then approximately what number does it converge to?

Solution:

```
> S[1] := evalf(exp(-1)):
  for N from 2 to 20 do
  S[N] := evalf(S[N-1] + exp(-N)/N):
  end do:
> plot([seq([N,S[N]], N = 1 .. 20)], style = POINT);
```




```
> S[19], S[20];
```

$$0.4586751452, 0.4586751453$$

The graphical evidence suggests that the series converges to a number not too much bigger than .4586751453.

In fact, the series can be summed exactly:

```
> sum(exp(-n)/n,n=1..infinity);
```

$$-\ln\left(1-\frac{1}{e}\right)$$

```
> evalf( -ln(1-1/exp(1)) );
```

$$0.4586751454$$

Ω

Exercise 52. (page 630)

Plot the partial sums of $\sum_{n=1}^{\infty} \frac{\sqrt{n}}{n!}$. From your plot does it appear that the given series converges? If so, then approximately what number does it converge to?

Solution:

```
> S[1] := evalf(sqrt(1)/1!):
  for N from 2 to 20 do
  S[N] := evalf(S[N-1] + sqrt(N)/N!):
  end do:
> plot([seq([N,S[N]], N = 1 .. 20)], style = POINT);
```




```
> S[19], S[20];
```

$$2.101755547, 2.101755547$$

The graphical evidence suggests that the series converges to a number not too much bigger than 2.101755547.

Maple cannot sum this series exactly, but its numerical evaluation is in line with our evidence:

```
> evalf(sum(sqrt(n)/n!,n = 1 .. infinity));
```

$$2.101755548$$

Ω

Exercise 53. (page 630)

Plot the partial sums of $\sum_{n=1}^{\infty} \left(\frac{1.1}{n}\right)^n$. From your plot does it appear that the given series converges? If so, then approximately what number does it converge to?

Solution:

```
> S[1] := 1.1:
  for N from 2 to 20 do
  S[N] := evalf(S[N-1] + (1.1/N)^N):
  end do:
> plot([seq([N,S[N]], N = 1 .. 20)], style = POINT);
```




```
> S[19], S[20];
```

$$1.458071271, 1.458071271$$

The graphical evidence suggests that the series converges to a number not too much bigger than 1.458071271.

Maple cannot sum this series exactly, but its numerical evaluation is in line with our evidence:

```
> evalf(sum((1.1/n)^n,n = 1 .. infinity));
```

$$1.458071271$$

Ω

Exercise 54. (page 630)

Plot the partial sums of $\sum_{n=1}^{\infty} \frac{\sin(n)}{n^3}$. From your plot does it appear that the given series converges? If so, then approximately what number does it converge to?

Solution:

```
> S[1] := sin(1.0):
  for N from 2 to 50 do
  S[N] := evalf(S[N-1] + sin(N)/N^3):
  end do:
> plot([seq([N,S[N]], N = 1 .. 50)], style = POINT);
```




```
> S[49], S[50];
```

$$0.9428633932, 0.9428612942$$

The graphical evidence suggests that the series converges to about 0.94286.

Maple cannot sum this series exactly and does not return a numerical evaluation**:**

```
> evalf(sum(sin(n)/n^3,n = 1 .. infinity));
```

$$\sum_{n=1}^{\infty} \frac{\sin(n)}{n^3}$$

Ω

Exercise 55. (page 630)

The geometric series $\sum_{n=1}^{\infty} \left(\frac{1}{1+10^{(-50)}}\right)^n$ converges, but its sum S is very large**.** By about how much does the millionth partial sum $S_{1000000}$ differ from the full sum S? How large must N be so that S_N is within 0.1 of S ?

Solution:

Let $r = \frac{1}{1+10^{(-50)}}$**.**

The difference between the full sum and the partial sum S_{10^6} is $\sum_{n=10^6+1}^{\infty} r^n$, or $\frac{r^{(10^6+1)}}{1-r}$**.**

We will write this difference as $r = \frac{1}{1+a^{(-b)}}$ and fill in $a = 10$ and $b = 50$ later**.**

```
> r := 1/(1+a^(-b));
```

$$r := \frac{1}{1+a^{(-b)}}$$

The difference ε is then given by $\varepsilon = \frac{r^c}{1-r}$ where $c = 10^6 + 1$ will be filled in later**.**

```
> epsilon := r^c/(1-r);
```

$$\varepsilon := \frac{\left(\frac{1}{1+a^{(-b)}}\right)^c}{1 - \frac{1}{1+a^{(-b)}}}$$

After some algebraic simplification we find $\varepsilon = \frac{a^{(b\,c)}}{(a^b+1)^{(c-1)}}$. Let us test this equation with **Maple**:

```
> testeq( epsilon  =  a^(b*c)/(a^b+1)^(c-1)  );
```

$$true$$

Set $E = \ln(\varepsilon)$ in order to facilitate working with very large numbers

```
> E := ln( a^(b*c)/(a^b+1)^(c-1) );
```

$$E := \ln\left(\frac{a^{(b\,c)}}{(a^b+1)^{(c-1)}}\right)$$

Simplifying the logarithm we find $E = b\,c\ln(a) - (c-1)\ln(a^b+1)$:

```
> e := b*c*ln(a)-(c-1)*ln(a^b+1);
```

$$e := b\,c\ln(a) - (c-1)\ln(a^b+1)$$

Next we numerically evaluate the logarithm of the error

```
> log_difference := evalf( subs({a=10, b=50, c = 10^6+1} , e),
  100);
```

$$log\_difference := 115.12925464970228420089957273421821038005507442143864880166\backslash$$
$$39504837863048386762401179986025 45$$

Thus, the difference $\mathbf{e}^{log\_difference}$ between the full sum S and the millionth partial sum $S_{1000000}$ is about

```
> exp(log_difference);
```

$$0.9999999503\ 10^{50}$$

After adding a million terms of the series we are still about 10^{50} away from the full sum.

Next we consider how large N must be so that S_N is within 0.1 of S.
First set δ to be the natural logarithm of the error**:**

```
> delta := ln(0.1);
```

$$\delta := -2.302585093$$

The natural logarithm of the difference between the full sum and the N^{th} partial sum is:

```
> Delta := subs({a=10, b=50, c = N+1}, e);
```

$$\Delta := 50\,(N+1)\ln(10) - N\ln(100000000000000000000000000000000000000000000000001)$$

We set $\Delta = \delta$ and solve for N**.** In order to facilitate the numerical root-finding algorithm, we first run a loop to determine a reasonable interval in which to search.

```
> for j from 50 to 60 do
  evalf(subs(N=10^j, Delta - delta),100), `N`=j;
  end do;
```

$$116.4318397427022842008995727342182103800550744314 4, N = 50$$

$$107.43183974270228420089957273421821038005507443 14, N = 51$$

$$17.431839742702284200899572734218210380055074431, N = 52$$

$$-882.5681602572977157991004272657817896199449255 7, N = 53$$

$$-9882.568160257297715799100427265781789619944925 6, N = 54$$

$$-99882.56816025729771579910042726578178961994492 6, N = 55$$

$$-999882.5681602572977157991004272657817896199449 3, N = 56$$

$$-0.99998825681602572977157991004272657817896199449\ 10^{7}, N = 57$$

$$-0.99999882568160257297715799100427265781789619945\ 10^{8}, N = 58$$

$$-0.99999988256816025729771579910042726578178961994\ 10^{9}, N = 59$$

$$-0.99999998825681602572977157991004272657817896199\ 10^{10}, N = 60$$

We should search in the interval $[10^{52}, 10^{53}]$. Let us narrow it down futher

```
> for j from 1 to 10 do
  evalf(subs(N=j*10^52, Delta - delta),100), `N`=j*10^52;
  end do;
```

17.4318397427022842008995727342182103800550744 31,

$N = 10000000000000000000000000000000000000000000000000000$

-82.56816025729771579910042726578178961994492 5569,

$N = 20000000000000000000000000000000000000000000000000000$

-182.5681602572977157991004272657817896199449 25569,

$N = 30000000000000000000000000000000000000000000000000000$

-282.5681602572977157991004272657817896199449 25569,

$N = 40000000000000000000000000000000000000000000000000000$

-382.5681602572977157991004272657817896199449 25569,

$N = 50000000000000000000000000000000000000000000000000000$

-482.5681602572977157991004272657817896199449 25569,

$N = 60000000000000000000000000000000000000000000000000000$

-582.5681602572977157991004272657817896199449 25569,

$N = 70000000000000000000000000000000000000000000000000000$

-682.5681602572977157991004272657817896199449 25569,

$N = 80000000000000000000000000000000000000000000000000000$

-782.5681602572977157991004272657817896199449 2557,

$N = 90000000000000000000000000000000000000000000000000000$

-882.5681602572977157991004272657817896199449 2557,

$N = 100000000000000000000000000000000000000000000000000000$

We should search in the interval $[10^{52}, 2\ 10^{52}]$. Let us narrow it down futher

```
> Digits := 70;
  fsolve( Delta = delta, N, 10^52 .. 2*10^52);
```

$Digits := 70$

$0.1174318397427022842008995727342182103800550744314386488016663950483786\ 10^{53}$

This figure represents the number of terms needed for an error no greater than 0.1 .

Ω

Code for The Figures

Copyright and Author Information

Worksheet Title: BlankKrantz-09_1R8.mws   A Maple Release 8 worksheet.

Author: Brian E. Blank
Date Created: 6 November 2007
Date Last Revised: 11 November 2007

For more information, please contact the author:

Department of Mathematics,
Washington University in St. Louis
St. Louis, MO 63130

Telephone: (314) 935-6763
e-mail: brian@math.wustl.edu